Лабораторная работа №1

Математические модели динамики численности популяции одного вида

Мат. моделирование динамических процессов 1
БГУ, ММФ, 3 курс, 6 семестр
специальность Компьютерная математика и системный анализ
февраль 2022
доц. Лаврова О.А., доц. Щеглова Н.Л.

Задание 1. Модель Мальтуса

Модель Мальтуса, или модель экспоненциального роста (1798) является одной из простейших моделей численности популяции одного вида. Модель была предложена Мальтусом для описания роста численности населения Земли.

Содержательная постановка задачи

Рассмотрим популяцию больших размеров, численность которой измеряется миллионами и более. Полагаем, что не существует факторов, сдерживающих рост популяции (болезни, хищники, конкурирующие виды, ограниченность питания). Игнорируем процессы иммиграции и эмиграции. Необходимо определить изменение численности популяции во времени.

Концептуальная постановка задачи

Численность популяции рассматривается как непрерывная функция времени $N = N(t)$. В силу того, что популяция достаточно велика, игнорируем случайные различия между отдельными особями. Полагаем, что каждая особь в популяции имеет равные шансы родить и умереть в течение определенного промежутка времени. Полагаем, что рождаемость и смертность непрерывны во времени и заданы коэффициентами рождаемости $\alpha(t) \geq 0$ и смертности $\beta(t) \geq 0$ на одну особь в единицу времени.

В основу модели Мальтуса положено утверждение, что скорость изменения численности популяции $N'(t)$ пропорциональна ее текущей численности $N(t)$ в момент времени t, с коэффициентом пропорциональности, равным разности коэффициентов рождаемости и смертности

$$N'(t) = (\alpha(t) - \beta(t))\, N(t). \tag{1}$$

Предположение вида “скорость изменения величины пропорциональна значению самой величины (или некоторой функции от нее)” широко используется в различных областях знаний.

Модель Мальтуса является примером применения принципа аналогии при построении математических моделей объектов, для которых невозможно указать фундаментальные законы или вариационные принципы, которым они подчиняются.

Математическая модель

Моделью Мальтуса называется задача Коши для уравнения (1) при заданном начальном условии в момент времени $t = t_0$

$$N'(t) = (\alpha(t) - \beta(t))\, N(t), \quad N(t_0) = N_0. \qquad (2)$$

Задание 1.1 (аналитическое решение)

Постройте аналитическое решение модели Мальтуса (2), например, методом разделения переменных для заданной численности населения N_0 в начальный момент времени t_0. Сравните построенное решение и решение, полученное с помощью функции **DSolve**

```
In[1]:= ClearAll[t0, N0];

In[2]:= sol = DSolveValue[{NN'[t] == (α[t] - β[t]) NN[t], NN[t0] == N0}, NN[t], t]
```

Out[2]= $e^{\int_1^t (\alpha[K[1]]-\beta[K[1]])\, dK[1] - \int_1^{t0} (\alpha[K[1]]-\beta[K[1]])\, dK[1]}\, N0$

Напишите пользовательскую функцию для задания аналитического решения модели Мальтуса, где $\alpha(t)$, $\beta(t)$ и $N_0 \geq 0$ и $t_0 \geq 0$ являются аргументами функции.

Задание 1.2 (график аналитического решения)

Изобразите в одной системе координат графики функции $N(t)$ для фиксированных значений t_0 и $N_0 > 0$ и различных значений $\alpha(t) = \alpha_0 = \text{const}$ и $\beta(t) = \beta_0 = \text{const}$. Рассмотрите случаи $\alpha_0 < \beta_0$, $\alpha_0 = \beta_0$, $\alpha_0 > \beta_0$.

Пример построения графиков решений

```
In[3]:= t0 = 1; N0 = 10;

In[4]:= funcs = (Activate@(sol /. #)) & /@
          {{α[_] → 0, β[_] → 2}, {α[_] → 2, β[_] → 2}, {α[_] → 2, β[_] → 0}}
```

Out[4]= $\{10\, e^{-2 (-1+t)},\ 10,\ 10\, e^{2 (-1+t)}\}$

In[5]:=
```
Plot[funcs, {t, t0, 3}, PlotLegends → {"α0<β0", "α0=β0", "α0>β0"}]
```

Out[5]=



При выполнении задания необходимо использовать пользовательскую функцию из Задания 1.1 для описания аналитического решения модели Мальтуса.

Задание 1.3 (качественный анализ модели по решению)

На основании построенных графиков в Задании 1.2 **сформулируйте** выводы о поведении решения модели Мальтуса при $t \to \infty$ для случаев $\alpha_0 < \beta_0$, $\alpha_0 = \beta_0$, $\alpha_0 > \beta_0$ (например, монотонное возрастание/убывание, экспоненциальное возрастание/убывание, колебания, экспоненциальный рост, гиперболический рост). **Определите**, является ли положение равновесия $N(t) \equiv N_0$, соответствующее решению модели при совпадении коэффициентов рождаемости и смертности $\alpha_0 = \beta_0$, устойчивым относительно изменения значений параметров α_0, β_0?

Задание 1.4 (качественный анализ модели по фазовому портрету)

Фазовое пространство для рассматриваемой динамической системы первого порядка является одномерным и представляет собой прямую линию ON, которая ограничена значениями $N \geq 0$ в силу того, что $N(t)$ является численностью населения.

Изобразите фазовые траектории (например, с помощью функции **StreamPlot**) решений динамической системы для случая постоянных коэффициентов $\alpha(t) = \alpha_0 = \text{const}$ и $\beta(t) = \beta_0 = \text{const}$ при $\alpha_0 < \beta_0$, $\alpha_0 = \beta_0$, $\alpha_0 > \beta_0$. Пользуясь определением устойчивости в фазовом пространстве **выясните**, являются ли положения равновесия динамической системы устойчивыми при условии, что $N(t) \geq 0$.

Например,

In[6]:=
```
α = 1;  β = 2;
```

In[7]:=
```
StreamPlot[{(α - β) x, 0}, {x, 0, 3}, {y, 0, .1},
 AspectRatio → Automatic, ImageSize → {{500, 20}}]
```

Out[7]=



Задание 2. Логистическая модель или модель Ферхюльста

Содержательная постановка задачи

Рассмотрим популяцию больших размеров, численность которой измеряется миллионами и более. Полагаем, что сдерживающим фактором роста популяции является ограниченность ресурсов. Игнорируем процессы иммиграции и эмиграции. Необходимо определить изменение численности популяции во времени.

Концептуальная постановка задачи и математическая модель

Логистическая модель или модель Ферхюльста (1838, 1845) принимает во внимание ограниченность доступных популяции ресурсов. В частности, считается, что существует равновесная численность популяции $N_p = \text{const} > 0$**,** которую может обеспечить окружающая среда, т.е. $N(t) \to N_p$ при $t \to \infty$. Это свойство называется эффектом насыщения численности населения.

При моделировании полагается также, что скорость изменения численности популяции пропорциональна самой численности, умноженной (в отличие от модели Мальтуса) на величину ее отклонения от равновесного значения N_p

$$N'(t) = N(t)\,k\left(1 - \frac{N(t)}{N_p}\right), \qquad k = \text{const} > 0. \tag{3}$$

Логистической моделью называется задача Коши для уравнения (3) при заданном начальном условии в момент времени $t = t_0$

$$\begin{aligned} &N'(t) = N(t)\,k\left(1 - \frac{N(t)}{N_p}\right), \\ &N(t_0) = N_0. \end{aligned} \tag{4}$$

Предположение о механизмах насыщения используются при построении многих моделей в различных областях знаний.

Существенным недостатком математической модели является тот факт, что равновесная численность популяции N_p вводится в качестве известного параметра, в то время как нахождение этой величины нередко является основной задачей исследования.

Современные прогнозы показывают, что численность человечества в обозримом будущем стабилизируется на уровне $N_p = 12 * 10^9$ человек. Такой же прогноз дается ООН.

Задание 2.1 (аналитическое решение)

Постройте аналитическое решение логистической модели (4) (например, методом разделения переменных) для заданной численности населения N_0 в начальный момент времени t_0.

Сравните построенное решение и решение, полученное с помощью функции **DSolve**.
Напишите пользовательскую функцию для задания аналитического решения логистической модели, где $k > 0$, $N_p > 0$, $N_0 \geq 0$ и $t_0 \geq 0$ являются аргументами функции.

Задание 2.2 (график аналитического решения)

Изобразите в одной системе координат графики функции $N(t)$ для фиксированных значений k, N_p и t_0 и различных значений N_0. Рассмотрите случаи, когда $N_0 < N_p$, $N_0 = N_p$, $N_0 > N_p$. При выполнении задания необходимо использовать пользовательскую функцию из Задания 2.1 для описания аналитического решения логистической модели.

Задание 2.3 (качественный анализ модели по решению)

На основании построенных графиков в Задании 2.2 **сформулируйте** выводы о поведении решения логистической модели при $t \to \infty$ для случаев $N_0 < N_p$, $N_0 = N_p$, $N_0 > N_p$ при $k > 0$.
Определите, является ли положение равновесия $N(t) \equiv N_p$, соответствующее решению модели при $N_0 = N_p$, устойчивым по начальным данным?

Задание 2.4 (качественный анализ модели по фазовому портрету)

Фазовое пространство для рассматриваемой динамической системы первого порядка является одномерным и представляет собой прямую линию ON, которая ограничена значениями $N \geq 0$ в силу того, что $N(t)$ является численностью населения.

Изобразите фазовые траектории решений динамической системы, например, с помощью функции **StreamPlot**. Пользуясь определением устойчивости в фазовом пространстве **выясните**, являются ли положения равновесия динамической системы устойчивыми при условии, что $k > 0$, $N(t) \geq 0$.

Задание 3. Нелинейный аналог модели Мальтуса

Математическая модель

Полагаем, что коэффициент рождаемости пропорционален численности населения (например, потому что особи популяции заинтересованы в ее росте) $\alpha(N) = \alpha_0 N$, $\alpha_0 = \text{const} > 0$, а коэффициент смертности постоянный $\beta(t) = \beta_0 = \text{const} > 0$. Тогда уравнение Мальтуса (1) преобразуется к виду

$$N'(t) = N(t)\,(\alpha_0 N(t) - \beta_0) \tag{5}$$

с квадратичной нелинейностью в правой части.

Математической моделью называется задача Коши для уравнения (5) при заданном начальном условии в момент времени $t = t_0$

$$\begin{aligned} &N'(t) = N(t)\,(\alpha_0 N(t) - \beta_0),\\ &N(t_0) = N_0. \end{aligned} \tag{6}$$

Задание 3.1 (аналитическое решение)

Постройте аналитическое решение нелинейной модели (6) для заданной численности населения N_0 в начальный момент времени t_0. При построении аналитического решения необходимо отдельно рассматривать случаи $N_0 < N_{кр}$, $N_0 = N_{кр}$, $N_0 > N_{кр}$, где $N_{кр} = \frac{\beta_0}{\alpha_0}$. При $N_0 > N_{кр}$ решение строится только на отрезке $[t_0, t^*]$, где t^* определяется из условия $\lim_{t \to t^*} N[t] = +\infty$.
Сравните построенное решение и решение, полученное с помощью функции **DSolve**.
Напишите пользовательскую функцию для задания аналитического решения нелинейной модели Мальтуса, где $N_{кр} > 0$, $\beta_0 > 0$, $N_0 \geq 0$ и $t_0 \geq 0$ являются аргументами функции.

Задание 3.2 (график аналитического решения)

Изобразите в одной системе координат графики функции $N(t)$ для фиксированных значений α_0 и β_0 и различных значений N_0. Рассмотрите случаи $N_0 < N_{кр}$, $N_0 = N_{кр}$, $N_0 > N_{кр}$, где $N_{кр} = \frac{\beta_0}{\alpha_0}$.
При выполнении задания необходимо использовать пользовательскую функцию из Задания 3.1 для описания аналитического решения нелинейной модели Мальтуса.

Задание 3.3 (качественный анализ модели по решению)

На основании построенных графиков в Задании 3.2 **сформулируйте** выводы о поведении решения нелинейной модели при $t \to \infty$ для случаев $N_0 < N_{кр}$, $N_0 = N_{кр}$, $N_0 > N_{кр}$, $N_{кр} = \frac{\beta_0}{\alpha_0}$.
Определите, является ли положение равновесия $N(t) \equiv N_{кр}$, соответствующее решению модели при $N_0 = N_{кр}$, устойчивым по начальным данным?

Задание 3.4 (качественный анализ модели по фазовому портрету)

Фазовое пространство для рассматриваемой динамической системы первого порядка является одномерным и представляет собой прямую линию ON, которая ограничена значениями $N \geq 0$ в силу того, что $N(t)$ является численностью населения.

Изобразите фазовые траектории решений динамической системы, например, с помощью функции **StreamPlot**. Пользуясь определением устойчивости в фазовом пространстве **выясните**, являются ли положения равновесия динамической системы устойчивыми при условии $N(t) \geq 0$, $\alpha_0 > 0$, $\beta_0 > 0$.

Задание 4. Предсказание численности населения по заданным параметрам модели

На основе нелинейной модели из Задания 3 **сделайте предсказание** о численности населения Беларуси через 100 лет. Значения параметров модели α_0, β_0, N_0, t_0 определите из базы знаний Wolfram Alpha. Коэффициент рождаемости α_0, полученный из базы знаний Wolfram Alpha, должен быть изменен для использования в рамках нелинейной модели

Мальтуса: $\alpha_{0,\,\text{nonlinear}} = \alpha_0 / N_0$

Сравните с аналогичными предсказаниями модели Мальтуса с постоянными коэффициентами, а также с данными о численности населения Беларуси из базы знаний Wolfram Alpha.

Некоторые рекомендации по работе с Wolfram Alpha

Создадим объект типа “Country”, ассоциированный с данными для имени “Belarus”

```
In[8]:= entityBelarus = Entity["Country", "Belarus"];
```

Получим необходимые свойства объекта EntityBelarus и уберем дополнительно размерности величин

```
In[9]:= {N0, α0, β0} = QuantityMagnitude@EntityValue[entityBelarus, #] & /@
          {"Population", "BirthRateFraction", "DeathRateFraction"}
Out[9]= {9 449 321, 0.011874, 0.013346}
```

Определим дату, которой соответствуют полученные данные

```
In[10]:= EntityValue[entityBelarus, #, "Date"] & /@
          {"Population", "BirthRateFraction", "DeathRateFraction"}
Out[10]= {Year: 2020, Year: 2017, Year: 2017}
```

```
In[11]:= t0 = 2017;
```

Скорректируем значение начальной популяции

```
In[12]:= N0 = QuantityMagnitude@Dated[entityBelarus, 2017]["Population"]
Out[12]= 9 450 233
```

Построим график функции для всех значений свойства “Population”, которые хранятся в базе знаний

```
In[13]:= ts = Dated[entityBelarus, All]["Population"];
In[14]:= DateListPlot[ts, PlotRange → {{"1950", "2050"}, All}, Joined → False]
```

Out[14]=



С помощью функций **TimeSeriesModelFit** и **TimeSeriesForecast** сделаем предсказание на следующие 100 лет по имеющимся данным из базы знаний WolframAlpha

In[15]:=
```
tsModel = TimeSeriesModelFit[ts];
tsForecast = TimeSeriesForecast[tsModel, {100}];
```

TemporalData: The data is not uniformly spaced and will be automatically resampled to the resolution of the minimum time increment.

In[17]:=
```
DateListPlot[{ts, tsForecast}, PlotRange → {{"1950", "2150"}, All}, Joined → False]
```

Out[17]=



In[18]:=
```
tsForecast["SliceData", "2122"]
```

Out[18]= $\{6.24397 \times 10^6$ people$\}$

Задание 5. Изменение численности населения страны. Определение параметров модели по реальным данным

По данным о численности населения для заданной страны в период с 1950 года до 2010 года **найдите** значения неизвестных параметров для
- модели Мальтуса с постоянными коэффициентами (см. Задание 1),
- логистической модели (см. Задание 2) и
- нелинейного аналоги модели Мальтуса (см. Задание 3),
полагая, что $t_0 = 1950$ и N_0 являются известными. Для этого необходимо аппроксимировать реальные данные функциями аналитических решений для указанных моделей с помощью функции **FindFit**.
Постройте графики относительной ошибки для найденных модельных решений в сравнении с реальными данными о численности населения.

Страны по вариантам:

1. Россия
2. Китай
3. Япония
4. Швеция
5. Франция
6. Греция

7. Марокко
8. Великобритания
9. Германия
10. Турция
11. Нидерланды
12. Бельгия
13. Катар
14. Египет

Рекомендации по реализации

Воспользуемся данными из базы знаний Wolfram Alpha для Беларуси

In[19]:=
```
data = {#, QuantityMagnitude@Dated[entityBelarus, #]["Population"]} & /@
    Range[1950, 2010];
```

In[20]:=
```
ListPlot[data]
```

Out[20]=



Параметры модели t_0 и N_0 полагаем заданными

In[21]:=
```
{t0, N0} = First@data
```

Out[21]= {1950, 7 722 155}

Найдем неизвестные параметры для модели Мальтуса с использованием функции **FindFit**

In[22]:=
```
ClearAll[k, t]
```

In[23]:=
```
solutionMalthus = N0 Exp[k (t - t0)];
```

In[24]:=
```
FindFit[data, solutionMalthus , {k}, t]
```

Out[24]= {k → 0.00523589}

Построим аналитическое решение для модели Мальтуса

In[25]:=
```
solutionMalthus = solutionMalthus /. FindFit[data, solutionMalthus , {k}, t]
```

Out[25]= $7\,722\,155\, e^{0.00523589\,(-1950+t)}$

Построим график популяционной динамики на основании модели Мальтуса

In[26]:=
```
Show[Plot[solutionMalthus, {t, 1950, 2010}], ListPlot[data]]
```

Out[26]=


Построим график относительной ошибки

In[27]:=
```
errorMalthus = (Abs[Last@# - (solutionMalthus /. t → First@#)] / Last@#) & /@ data;
```

In[28]:=
```
{Norm[errorMalthus], Norm[errorMalthus, Infinity]}
```

Out[28]=
```
{0.39922, 0.122267}
```

In[29]:=
```
ListPlot[errorMalthus]
```

Out[29]=


Задание 6. Рост численности населения Земли. Определение параметров модели по реальным данным

По данным о численности населения Земли, доступным из базы знаний Wolfram Alpha, **найдите** значения неизвестных параметров $N_{кр} > 0$ и $\beta_0 > 0$ для нелинейного аналога модели Мальтуса (см. Задание 3), полагая, что t_0 и N_0 являются заданными. Для нахождения неизвестных параметров модели необходимо аппроксимировать реальные данные функцией аналитического решения с помощью функции **FindFit**, см. Задание 5.

В рамках нелинейного аналога модели Мальтуса **найдите** момент времени t^*, при котором численность населения Земли примет бесконечное значение, т.е. $\lim_{t \to t^*} N[t] = +\infty$, см. Задание 3.1.

Изобразите данные о численности населения Земли из базы знаний Wolfram Alpha, а также модельное решение на временном отрезке $[t_0, t^*]$.

Постройте график относительной ошибки для найденного модельного решения в сравнении с реальными данными о численности населения Земли, см. Задание 5.

Рекомендации по реализации

Воспользуемся данными из базы знаний Wolfram Alpha для населения Земли

In[30]:=
```
entityWorld = Entity["Country", "World"];
```

In[31]:=
```
dataWorld = Dated[entityWorld, All]["Population"];
```

In[32]:=
```
DateListPlot[dataWorld, Joined→False]
```

Out[32]=



Параметры модели t_0 и N_0 полагаем заданными

In[33]:=
```
{t0, N0} = {dataWorld["FirstDate"], dataWorld["FirstValue"]}
```

Out[33]= { Mon 1 Jan −10000 00:00:00 GMT+3. , $1. \times 10^6$ people }

Задание 7* (необязательное). Математическая модель радиоактивного распада

Постройте математическую модель радиоактивного распада вещества по аналогии с моделью Мальтуса. Сделайте предположение, что скорость распада радиоактивного вещества пропорционален количеству этого вещества $x = x(t)$, где коэффициент пропорциональности является постоянным $k = \text{const} < 0$. Для информации смотрите [5, с. 18].

Используя вид аналитического решения математической модели в виде задачи Коши с начальным условием $x(t) = x_0$ определите, как связан коэффициент пропорциональности k со временем T полураспада радиоактивного вещества: $x(T) = x_0 / 2$?

Дополнительно решите задачу 2.7 и задачу 2.8 из [5, с.18].

Задание 8* (необязательное). Математическая модель распространения нового продукта на рынке

Данное задание является примером построения и анализа математической модели в экономике.

Постройте математическую модель распространения нового продукта или услуги на рынке (модель Басса). В качестве неизвестной величины рассматривается количество пользователей некоторого продукта или услуги $N(t)$ на рынке. Используя принцип аналогии полагается, что за счет рекламы скорость изменения пользователей $N'(t)$ пропорциональна числу потенциальных клиентов $N_0 - N(t)$, где N_0 -- общее количество людей на рынке. Дополнительно учитывается эффект косвенной рекламы за счет общения пользователей продукта с потенциальными клиентами. Для информации смотрите [1, с. 150].

Анализируя математическую модель, определите момент времени, начиная с которого продолжать рекламу становится невыгодным. Для информации смотрите [1, с. 150].

Литература

[1] А. А. Самарский, А. П. Михайлов. Математическое моделирование: Идеи. Методы. Примеры. -- М.: Физматлит, 2001.

[2] А. С. Братусь, А. С. Новожилов, А. П. Платонов. Динамические системы и модели биологии. -- 2011.

[3] Д. Мюррей. Математическая биология. Том I. Введение. -- М.-Ижевск, 2009.

[4] B. Barnes and G. R. Fulford. Mathematical Modelling with Case Studies: A differential equation approach using Maple and MATLAB. Second Edition. -- CRC Press, 2008.

[5] Р. А. Прохорова. Обыкновенные дифференциальные уравнения : учебное пособие. -- Мн. : БГУ, 2017. http://elib.bsu.by/handle/123456789/205697

[6] В. В. Амелькин. Дифференциальные уравнения : учебное пособие. -- Мн. : БГУ, 2012. http://elib.bsu.by/handle/123456789/43871